Stata 12/13 **Tutorial 3**

TOPIC: **Modeling Marginal Effects of Continuous Explanatory Variables: Constant or Variable?**

DATA: **auto1.dta** (a *Stata*-format dataset first created in ***Stata 12/13 Tutorial 1***)

TASKS: ***Stata 12/13 Tutorial 3*** examines two alternative specifications of a linear regression model containing two *continuous* explanatory variables: one restricts the marginal effects of the two explanatory variables to be constants; the other allows the marginal effects of the two explanatory variables to be variable. The first specification was explored in some depth in *Stata 12/13 Tutorial 2*. The second specification of the model introduces variable marginal effects by including as regressors ***polynomial terms*** and ***interaction terms*** in the two continuous explanatory variables. A third model is used to investigate the question of how many polynomial terms are needed to adequately represent the effect of the two continuous explanatory variables on the dependent variable.

- The ***Stata* commands** that constitute the primary subject of this tutorial are:

regress	Used to perform OLS estimation of multiple linear regression models.
test	Used to compute F-tests of linear coefficient equality restrictions after OLS estimation.
lincom	Used after estimation to compute linear combinations of coefficient estimates and associated statistics.

- The ***Stata* statistical functions** used in this tutorial are:

ttail(df, t_0**)**	Computes ***right-tail p-values of calculated t-statistics***
invttail(df, p**)**	Computes ***right-tail critical values of t-distributions***
Ftail(df_1, df_2, F_0**)**	Computes ***p-values of calculated F-statistics***
invFtail(df_1, df_2, α**)**	Computes ***critical values of F-distributions***

NOTE: *Stata* commands are *case sensitive*. All *Stata command names* must be typed in the Command window in ***lower case* letters**.

❑ Preparing for Your *Stata* Session

Before beginning your *Stata* session, use Windows Explorer to copy the *Stata*-format data set **auto1.dta** to the *Stata* ***working directory*** on the C:-drive or D:-drive of the computer at which you are working.

- **On the computers in Dunning 350**, the default *Stata* working directory is usually **C:\data**.

- **On the computers in MC B111**, the default *Stata* working directory is usually **D:\courses**.

❑ Start Your *Stata* Session

To start your *Stata* session, double-click on the ***Stata 12* or *Stata 13* icon** on the Windows desktop.

After you double-click the ***Stata 12* or *Stata 13* icon**, you will see the familiar screen of five *Stata* windows.

❑ Record Your *Stata* Session – log using

To record your *Stata* session, including all the *Stata* commands you enter and the results (output) produced by these commands, make a text-format **.log** file named **452tutorial3.log**. To open (begin) the log file **452tutorial3.log**, enter in the Command window:

```
log using 452tutorial3.log
```

This command opens a text-format (ASCII) file called **452tutorial3.log** in the current *Stata* working directory.

Note: It is important to include the **.log** file extension when opening a log file; if you do not, your log file will be in smcl format, a format that only *Stata* can read. Once you have opened the **452tutorial3.log** file, a copy of all the commands you enter during your *Stata* session and of all the results they produce is recorded in that **452tutorial3.log** file.

An alternative way to open (start) a text-format log file is to use the **Log** button in the button bar near the top of the *Stata* window.

The following steps would replicate what the above **log using** command does.

- Click on the **Log** button in the button bar near the top of the *Stata* window; this opens the **Begin Logging Stata Output** dialog box.

- In the **Begin Logging Stata Output** dialog box:

  - click on **Save as type:** and select **`Log (*.log)`**;
  - click on the **File name:** box and type the file name **`452tutorial3`**;
  - click on the **Save** button.

❑ Load a *Stata*-Format Dataset into *Stata* – use

Load, or read, into memory the data set you are using. To load the *Stata*-format data file **auto1.dta** into memory, enter in the Command window:

```
use auto1
```

This command loads into memory the *Stata*-format data set **auto1.dta**.

❑ Familiarize Yourself with the Current Data Set

To familiarize (or re-familiarize) yourself with the contents of the current data set, type in the Command window the following commands:

```
describe, short
describe
summarize
list price wgt mpg
```

❑ A Linear Regression Model with Constant Marginal Effects

Nature: A linear regression equation in which the continuous explanatory variables enter only linearly exhibits constant marginal effects.

Consider the following generic linear regression equation in the two continuous explanatory variables X_1 and X_2:

$$Y_i = \beta_0 + \beta_1 X_{i1} + \beta_2 X_{i2} + u_i \,. \qquad (1)$$

Formally, the ***population regression function*** $E(Y_i \,|\, X_{i1}, X_{i2}) = f(X_{i1}, X_{i2})$ in PRE (1) can be derived as a first-order Taylor series approximation to the function $f(X_{i1}, X_{i2})$.

Compare the marginal effects on Y of the two explanatory variables X_1 and X_2 in equation (1).

- The **marginal effect of X_1** in equation (1) is:

$$\frac{\partial Y_i}{\partial X_{i1}} = \frac{\partial E\left(Y_i \,\middle|\, X_{i1}, X_{i2}\right)}{\partial X_{i1}} = \beta_1 = \text{a constant}$$

- The **marginal effect of X_2** in equation (1) is:

$$\frac{\partial Y_i}{\partial X_{i2}} = \frac{\partial E\left(Y_i \,\middle|\, X_{i1}, X_{i2}\right)}{\partial X_{i2}} = \beta_2 = \text{a constant}$$

In other words, the slope coefficients of X_1 and X_2 in equation (1) equal the corresponding marginal effects of the two explanatory variables.

Example: An example of the generic regression model (1) is the following population regression equation for North American car prices:

Model 1:

$$\text{price}_i = \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + u_i \qquad \textbf{(2)}$$

where

- $price_i$ = the price of the i-th car (in US dollars);
- wgt_i = the weight of the i-th car (in pounds);
- mpg_i = the miles per gallon (fuel efficiency) for the i-th car (in miles per gallon).

- To estimate by OLS the linear regression model given by PRE (2), enter in the Command window the following **regress** command:

```
regress price wgt mpg
```

♦ ***Test 1:*** Test the hypothesis that the ***marginal* effect of *wgt_i* on *price_i* is *zero*** in regression equation (2). Use *Stata* commands to demonstrate that an F-test and a two-tail t-test of this hypothesis are equivalent.

◊ The **marginal effect of *wgt_i* on *price_i* in Model 1** is obtained by partially differentiating regression equation (2) with respect to wgt_i.

$$\frac{\partial\, price_i}{\partial\, wgt_i} = \frac{\partial\, E\left(price_i \mid wgt_i, mpg_i\right)}{\partial\, wgt_i} = \beta_1$$

- The ***null* and *alternative* hypotheses** are therefore:

H_0: $\beta_1 = 0$

H_1: $\beta_1 \neq 0$.

- The following **test** command computes an **F-test of H_0 against H_1**. The **return list** command displays the temporarily saved results of the **test** command. Enter the commands:

```
test wgt
return list
```

Inspect the results generated by these commands. State the inference you would draw from this test.

- Use a **scalar** command to save the sample value of the F-statistic for $\hat{\beta}_1$, denoted analytically as $F_0(\hat{\beta}_1)$, and display the value of $F_0(\hat{\beta}_1)$ and its p-value. Enter the commands:

```
scalar Fb1 = r(F)
scalar list Fb1
display Ftail(1, 71, Fb1)
```

- The following **lincom** command computes a ***two-tail* t-test of H_0 against H_1**. Enter the commands:

```
lincom _b[wgt]
return list
```

  Inspect the results generated by these commands. State the inference you would draw from this test.

- Use a **scalar** command to save the sample value of the t-statistic for $\hat{\beta}_1$, denoted analytically as $t_0(\hat{\beta}_1)$, and display the value of $t_0(\hat{\beta}_1)$ and its two-tail p-value (computed in two different ways). Enter the commands:

```
scalar tb1 = r(estimate)/r(se)
scalar list tb1
display 2*ttail(71, abs(tb1))
display 2*ttail(r(df), abs(r(estimate)/r(se)))
```

- Compare the results of the F-test and two-tail t-test of H_0 against H_1. This can be done in either of two equivalent ways. First, show that the sample values of the test statistics $F_0(\hat{\beta}_1)$ and $t_0(\hat{\beta}_1)$ are related as follows:

$$F_0(\hat{\beta}_1) = t_0(\hat{\beta}_1)^2.$$

  Since $F_0(\hat{\beta}_1)$ and $t_0(\hat{\beta}_1)^2$ have the same null distribution, namely the F[1,71] distribution, it follows that they have the same p-value. Enter the commands:

```
scalar tb1sq = tb1^2
scalar list Fb1 tb1sq
display Ftail(1, 71, Fb1)
display Ftail(1, 71, tb1sq)
```

You should be able to see from the results of these commands that the F-test and two-tail t-test of H_0: $\beta_1 = 0$ against H_1: $\beta_1 \neq 0$ are equivalent, meaning they yield identical inferences.

- A second way to demonstrate the equivalence of the F-test and two-tail t-test of H_0 against H_1 is to show that the sample values of the test statistics $F_0(\hat{\beta}_1)$ and $t_0(\hat{\beta}_1)$ are related as follows:

$$t_0(\hat{\beta}_1) = \sqrt{F_0(\hat{\beta}_1)}.$$

Since $t_0(\hat{\beta}_1)$ and $\sqrt{F_0(\hat{\beta}_1)}$ have the same null distribution, namely the t[71] distribution, it follows that they have the same p-value. Enter the commands:

```
scalar sqrtFb1 = sqrt(Fb1)
scalar list tb1 sqrtFb1
display 2*ttail(71, abs(tb1))
display 2*ttail(71, abs(sqrtFb1))
```

Again, you should be able to see from the results of these commands that the F-test and two-tail t-test of H_0: $\beta_1 = 0$ against H_1: $\beta_1 \neq 0$ are equivalent, meaning they yield identical inferences.

♦ ***Test 2:*** Test the hypothesis that the ***marginal* effect of *mpg*$_i$ on *price*$_i$ is *zero*** in regression equation (2). Use *Stata* commands to demonstrate that an F-test and a two-tail t-test of this hypothesis are equivalent.

◊ The **marginal effect of *mpg*$_i$ on *price*$_i$ in Model 1** is obtained by partially differentiating regression equation (2) with respect to mpg_i.

$$\frac{\partial\, price_i}{\partial\, mpg_i} = \frac{\partial\, E\left(price_i \mid wgt_i, mpg_i\right)}{\partial\, mpg_i} = \beta_2$$

- The ***null* and *alternative* hypotheses** are therefore:

  H_0: $\beta_2 = 0$

  H_1: $\beta_2 \neq 0$.

- The following **test** command computes an **F-test of H_0 against H_1**. The **return list** command displays the temporarily saved results of the **test** command. Enter the commands:

```
test mpg
return list
```

  What inference would you draw from the generated results of this test?

- Use a **scalar** command to save the sample value of the F-statistic for $\hat{\beta}_2$, denoted analytically as $F_0(\hat{\beta}_2)$, and display the value of $F_0(\hat{\beta}_2)$ and its p-value. Enter the commands:

```
scalar Fb2 = r(F)
scalar list Fb2
display Ftail(1, 71, Fb2)
```

- The following **lincom** command computes a ***two-tail* t-test of H_0 against H_1**. Enter the commands:

```
lincom _b[mpg]
return list
```

  Inspect the results generated by these commands. State the inference you would draw from this test.

- Use a **scalar** command to save the sample value of the t-statistic for $\hat{\beta}_2$, denoted analytically as $t_0(\hat{\beta}_2)$, and display the value of $t_0(\hat{\beta}_2)$ and its two-tail p-value. Enter the commands:

```
scalar tb2 = r(estimate)/r(se)
scalar list tb2
display 2*ttail(71, abs(tb2))
```

- Compare the results of the F-test and two-tail t-test of H_0 against H_1. This can be done in either of two equivalent ways. First, show that the sample values of the test statistics $F_0(\hat{\beta}_2)$ and $t_0(\hat{\beta}_2)$ are related as follows:

  $$F_0(\hat{\beta}_2) = t_0(\hat{\beta}_2)^2.$$

  Since $F_0(\hat{\beta}_2)$ and $t_0(\hat{\beta}_2)^2$ have the same null distribution, namely the F[1,71] distribution, it follows that they have the same p-value. Enter the commands:

  ```
  scalar tb2sq = tb2^2
  scalar list Fb2 tb2sq
  display Ftail(1, 71, Fb2)
  display Ftail(1, 71, tb2sq)
  ```

  You should be able to see from the results of these commands that the F-test and two-tail t-test of H_0: $\beta_2 = 0$ against H_1: $\beta_2 \neq 0$ are equivalent, meaning they yield identical inferences.

- A second way to demonstrate the equivalence of the F-test and two-tail t-test of H_0 against H_1 is to show that the sample values of the test statistics $F_0(\hat{\beta}_2)$ and $t_0(\hat{\beta}_2)$ are related as follows:

  $$t_0(\hat{\beta}_2) = \sqrt{F_0(\hat{\beta}_2)}.$$

  Since $t_0(\hat{\beta}_2)$ and $\sqrt{F_0(\hat{\beta}_2)}$ have the same null distribution, namely the t[71] distribution, it follows that they should have the same p-value. Enter the commands:

  ```
  scalar sqrtFb2 = sqrt(Fb2)
  scalar list tb2 sqrtFb2
  display 2*ttail(71, abs(tb2))
  display 2*ttail(71, abs(sqrtFb2))
  ```

  Again, you should be able to see from the results of these commands that the F-test and two-tail t-test of H_0: $\beta_2 = 0$ against H_1: $\beta_2 \neq 0$ are equivalent, meaning they yield identical inferences.

❑ Interaction Variables: Squares and Cross Products of Continuous Variables Yield Variable Marginal Effects

Nature: ***Interactions* between two *continuous* variables** refer to products of explanatory variables. For example, if X_{ij} and X_{ih} are two continuous explanatory variables, the interaction term between them is the product $X_{ij}X_{ih}$. Alternatively, the interaction of the variable X_{ij} with itself is simply the product $X_{ij}X_{ij} = X_{ij}^2$. Inclusion of these regressors in a linear regression model allows for nonconstant marginal effects of the explanatory variables on the conditional mean of the dependent variable Y.

Usage: ***Interaction terms*** between ***continuous* variables** allow the marginal effect of one explanatory variable to be a linear function of both itself and the other explanatory variable.

Consider the following generic linear regression equation:

$$Y_i = \beta_0 + \beta_1 X_{i1} + \beta_2 X_{i2} + \beta_3 X_{i1}^2 + \beta_4 X_{i2}^2 + \beta_5 X_{i1} X_{i2} + u_i . \qquad (3)$$

Formally, the population regression function $E(Y_i \mid X_{i1}, X_{i2}) = f(X_{i1}, X_{i2})$ in PRE (3) can be derived as a ***second-order*** Taylor series approximation to the function $f(X_{i1}, X_{i2})$.

Compare the marginal effects on Y of the two explanatory variables X_1 and X_2 in equation (3).

- The **marginal effect of X_1** in equation (3) is:

$$\frac{\partial Y_i}{\partial X_{i1}} = \frac{\partial E\left(Y_i \mid X_{i1}, X_{i2}\right)}{\partial X_{i1}} = \beta_1 + 2\beta_3 X_{i1} + \beta_5 X_{i2}$$

= a linear function of *both* X_{i1} and X_{i2}

- The **marginal effect of X_2** in equation (3) is:

$$\frac{\partial Y_i}{\partial X_{i2}} = \frac{\partial E\left(Y_i \mid X_{i1}, X_{i2}\right)}{\partial X_{i2}} = \beta_2 + 2\beta_4 X_{i2} + \beta_5 X_{i1}$$

$$= \text{ a linear function of } \textit{both } X_{i1} \text{ and } X_{i2}$$

Example: An example of the generic regression model (3) is the following population regression equation for North American car prices:

Model 2:

$$\text{price}_i = \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + u_i . \qquad \textbf{(4)}$$

where

price_i = the price of the i-th car (in US dollars);
wgt_i = the weight of the i-th car (in pounds);
wgt_i^2 = the square of wgt_i for the i-th car;
mpg_i = the miles per gallon (fuel efficiency) for the i-th car (in miles per gallon);
mpg_i^2 = the square of mpg_i for the i-th car;
wgt_impg_i = the product (interaction) of wgt_i and mpg_i for the i-th car.

- Before estimating regression equation (4), it is necessary to create the last three regressors in PRE (4), namely the squared terms wgt_i^2 and mpg_i^2, and the interaction term wgt_impg_i. Enter the following **generate** commands:

```
generate wgtsq = wgt*wgt
generate mpgsq = mpg^2
generate wgtmpg = wgt*mpg
```

Note the two equivalent ways of creating the square of a variable.

- Now estimate by OLS regression equation (4). Enter the following **regress** commands:

```
regress price wgt mpg wgtsq mpgsq wgtmpg
regress price wgt mpg wgtsq mpgsq wgtmpg, level(99)
```

How do the results of these two **regress** commands differ?

- ***Test 3:*** The first question to ask of Model 2 is whether the additional regressors it introduces into Model 1 – namely wgt_i^2, mpg_i^2 and $wgt_i mpg_i$ – are necessary in order to adequately represent the relationship of $price_i$ to the two continuous explanatory variables wgt_i and mpg_i.

- Compare Models 1 and 2.

Model 1:

$$price_i = \beta_0 + \beta_1 wgt_i + \beta_2 mpg_i + u_i \qquad \textbf{(2)}$$

Model 2:

$$price_i = \beta_0 + \beta_1 wgt_i + \beta_2 mpg_i + \beta_3 wgt_i^2 + \beta_4 mpg_i^2 + \beta_5 wgt_i mpg_i + u_i . \qquad \textbf{(4)}$$

It is evident from a comparison of Models 1 and 2 that Model 1 imposes on Model 2 three coefficient exclusion restrictions, namely the restrictions that $\beta_3 = 0$, $\beta_4 = 0$ and $\beta_5 = 0$. These are the three coefficient exclusion restrictions that we need to test.

- The ***null*** **and** ***alternative*** **hypotheses** are:

H_0: $\beta_3 = 0$ *and* $\beta_4 = 0$ *and* $\beta_5 = 0$

H_1: $\beta_3 \neq 0$ *and/or* $\beta_4 \neq 0$ *and/or* $\beta_5 \neq 0$.

Note that the null hypothesis H_0 implies that Model 1 is empirically adequate, whereas the alternative hypothesis H_1 implies rejection of Model 1 in favor of Model 2. As you will discover from later hypothesis tests in this section, the three coefficient exclusion restrictions specified by the null hypothesis H_0 imply that both the marginal effect of wgt_i on $price_i$ and the marginal effect of mpg_i on $price_i$ are constant, meaning they do not depend on wgt_i or mpg_i.

- The following **test** command computes an F-test of H_0 against H_1. Enter the commands:

```
test wgtsq mpgsq wgtmpg
return list
```

Inspect the results generated by these commands. State the inference you would draw from this test. Based on the results of this test, would you retain or reject Model 1 against Model 2?

♦ ***Test 4:*** Test the hypothesis that the ***marginal* effect of *wgt*$_i$ on *price*$_i$ is *zero*** in Model 2 given by regression equation (4).

- The marginal effect of wgt_i on $price_i$ is obtained by partially differentiating regression equation (4) with respect to wgt_i:

$$\frac{\partial\, price_i}{\partial\, wgt_i} = \beta_1 + 2\beta_3 wgt_i + \beta_5 mpg_i \,.$$

- Sufficient conditions for $\partial\, price_i / \partial\, wgt_i = 0$ for all i are $\beta_1 = 0$ *and* $\beta_3 = 0$ *and* $\beta_5 = 0$. We therefore want to test these three coefficient exclusion restrictions.

- The ***null* and *alternative* hypotheses** are:

  H_0: $\beta_1 = 0$ *and* $\beta_3 = 0$ *and* $\beta_5 = 0$

  H_1: $\beta_1 \neq 0$ *and/or* $\beta_3 \neq 0$ *and/or* $\beta_5 \neq 0$.

- The following **test** command computes an F-test of H_0 against H_1. Enter the commands:

```
test wgt wgtsq wgtmpg
return list
```

Inspect the results generated by these commands. State the inference you would draw from this test.

- Use a **lincom** command to compute the estimated value of the marginal effect of wgt_i in equation (4) for a car that weighs 2,500 pounds and gets 25 miles to the gallon. Recall that the expression for $\partial price_i / \partial wgt_i$ in equation (4) is:

$$\frac{\partial price_i}{\partial wgt_i} = \beta_1 + 2\beta_3 wgt_i + \beta_5 mpg_i .$$

Evaluated at $wgt_i = 2500$ and $mpg_i = 25$, the *estimated* marginal effect of wgt_i on $price_i$ is:

$$\text{estimate of } \frac{\partial price_i}{\partial wgt_i} = \hat{\beta}_1 + 2\hat{\beta}_3(2500) + \hat{\beta}_5(25) .$$

- Enter the commands:

```
lincom wgt + 2*wgtsq*2500 + wgtmpg*25
lincom _b[wgt] + 2*_b[wgtsq]*2500 + _b[wgtmpg]*25
```

Note that in the first **lincom** command *variable names* are used to represent their corresponding *coefficient estimates*. Inspect the output generated by these **lincom** commands (they are identical). Would you infer that the marginal effect of wgt_i on $price_i$ is different from zero for a car that weighs 2,500 pounds and has fuel efficiency equal to 25 miles per gallon?

♦ ***<u>Test 5</u>:*** Test the hypothesis that the ***marginal* effect of *mpg_i* on *$price_i$* is *zero*** in Model 2 given by regression equation (4).

- The marginal effect of mpg_i on $price_i$ is obtained by partially differentiating regression equation (4) with respect to mpg_i.

$$\frac{\partial price_i}{\partial mpg_i} = \beta_2 + 2\beta_4 mpg_i + \beta_5 wgt_i .$$

- Sufficient conditions for $\partial price_i / \partial mpg_i = 0$ for all i are $\beta_2 = 0$ *and* $\beta_4 = 0$ *and* $\beta_5 = 0$. We therefore want to test these three coefficient restrictions.

- The ***null*** **and** ***alternative*** **hypotheses** are:

  H_0: $\beta_2 = 0$ *and* $\beta_4 = 0$ *and* $\beta_5 = 0$

  H_1: $\beta_2 \neq 0$ *and/or* $\beta_4 \neq 0$ *and/or* $\beta_5 \neq 0$.

- The following **test** command computes an F-test of H_0 against H_1. Enter the commands:

```
test mpg mpgsq wgtmpg
return list
display Ftail(r(df), r(df_r), r(F))
```

  Inspect the results generated by these commands. State the inference you would draw from this test.

- Use a **lincom** command to compute the estimated value of the marginal effect of mpg_i in equation (4) for a car that weighs 2,500 pounds and gets 25 miles to the gallon. Recall that the expression for $\partial price_i / \partial mpg_i$ in equation (4) is:

$$\frac{\partial price_i}{\partial mpg_i} = \beta_2 + 2\beta_4 mpg_i + \beta_5 wgt_i .$$

  Evaluated at $wgt_i = 2500$ and $mpg_i = 25$, the *estimated* marginal effect of mpg_i on $price_i$ is:

$$\text{estimate of } \frac{\partial price_i}{\partial mpg_i} = \hat{\beta}_2 + 2\hat{\beta}_4(25) + \hat{\beta}_5(2500).$$

- Enter the commands:

```
lincom mpg + 2*mpgsq*25 + wgtmpg*2500
lincom _b[mpg] + 2*_b[mpgsq]*25 + _b[wgtmpg]*2500
```

  Note again that in the first of the above **lincom** commands *variable names* are used to represent their corresponding *coefficient estimates*. Would you infer that the marginal effect of mpg_i on $price_i$ is different from zero for a car that weighs 2,500 pounds and has fuel efficiency equal to 25 miles per gallon?

- ***<u>Test 6</u>:*** Test the hypothesis that the ***marginal* effect of *wgt_i* on *$price_i$* is *constant*** in Model 2 given by regression equation (4).

- Recall that the marginal effect of wgt_i on $price_i$ is obtained by partially differentiating regression equation (4) with respect to wgt_i:

$$\frac{\partial\, price_i}{\partial\, wgt_i} = \beta_1 + 2\beta_3 wgt_i + \beta_5 mpg_i .$$

- The marginal effect of wgt_i on $price_i$ is equal to the constant β_1 if $\beta_3 = 0$ *and* $\beta_5 = 0$. Hence sufficient conditions for $\partial\, price_i / \partial\, wgt_i = \beta_1$ = a constant for all i are $\beta_3 = 0$ *and* $\beta_5 = 0$. So we want to test these two coefficient exclusion restrictions.

- The ***null* and *alternative* hypotheses** are:

  H_0: $\beta_3 = 0$ *and* $\beta_5 = 0$

  H_1: $\beta_3 \neq 0$ *and/or* $\beta_5 \neq 0$.

- Since the above null hypothesis specifies ***two*** coefficient restrictions on regression equation (4), we must use an F-test. The following **test** command computes an F-test of H_0 against H_1. Enter the commands:

```
test wgtsq wgtmpg
return list
```

  Inspect the results generated by these commands. State the inference you would draw from this test.

- ***<u>Test 7</u>:*** Test the hypothesis that the ***marginal* effect of *mpg_i* on *$price_i$* is *constant*** in Model 2 given by regression equation (4).

- Recall that the marginal effect of mpg_i on $price_i$ is obtained by partially differentiating regression equation (4) with respect to mpg_i:

$$\frac{\partial\, price_i}{\partial\, mpg_i} = \beta_2 + 2\beta_4 mpg_i + \beta_5 wgt_i .$$

- The marginal effect of mpg_i on $price_i$ is equal to the constant β_2 if $\beta_4 = 0$ *and* $\beta_5 = 0$. Hence sufficient conditions for $\partial price_i / \partial mpg_i = \beta_2$ = a constant for all i are $\beta_4 = 0$ *and* $\beta_5 = 0$. We therefore want to test these two coefficient restrictions.

- The ***null*** **and *alternative* hypotheses** are:

  H_0: $\beta_4 = 0$ *and* $\beta_5 = 0$

  H_1: $\beta_4 \neq 0$ *and/or* $\beta_5 \neq 0$.

- Since the above null hypothesis specifies two coefficient restrictions on regression equation (4), we must use an F-test. The following **test** command computes an F-test of H_0 against H_1. Enter the commands:

```
test mpgsq wgtmpg
return list
```

  Inspect the results generated by these commands. State the inference you would draw from this test.

♦ ***<u>Test 8</u>:*** Test the hypothesis that the ***marginal* effect of *wgt*** wgt_i **on** $price_i$ **is unrelated to** wgt_i in Model 2 given by regression equation (4).

- Recall that the marginal effect of wgt_i on $price_i$ is obtained by partially differentiating regression equation (4) with respect to wgt_i:

$$\frac{\partial price_i}{\partial wgt_i} = \beta_1 + 2\beta_3 wgt_i + \beta_5 mpg_i .$$

- A sufficient condition for $\partial price_i / \partial wgt_i$ to be unrelated to wgt_i for all i is $\beta_3 = 0$. We therefore want to test this coefficient exclusion restriction on PRE (4).

- The ***null*** **and *alternative* hypotheses** are:

  H_0: $\beta_3 = 0$

  H_1: $\beta_3 \neq 0$.

- The following **test** command computes an **F-test** of H_0 against H_1. Enter the commands:

  ```
  test wgtsq    or    test wgtsq = 0
  return list
  ```

  Inspect the results generated by these commands. State the inference you would draw from this F-test.

- The following **lincom** command computes a ***two-tail*** **t-test** of H_0 against H_1. Enter the commands:

  ```
  lincom _b[wgtsq]
  return list
  scalar t8 = r(estimate)/r(se)
  scalar list t8
  display 2*ttail(68, abs(t8))
  display t8^2
  display Ftail(1, 68, t8^2)
  ```

  Inspect the results generated by these commands. State the inference you would draw from this t-test.

  Remarks: The above **lincom** command simply replicates the t-test of this same hypothesis that is reported by the **regress** command used to estimate regression equation (4) by OLS. You should be able to explain and verify from the above results why the **two-tail t-test** of H_0: $\beta_3 = 0$ versus H_1: $\beta_3 \neq 0$ ***is equivalent to*** the preceding **F-test** of the same hypothesis.

♦ ***Test 9:*** Consider the following two propositions:

1. The ***marginal*** **effect of** wgt_i **on** $price_i$ **is unrelated to** mpg_i in Model 2 given by regression equation (4).
2. The ***marginal*** **effect of** mpg_i **on** $price_i$ **is unrelated to** wgt_i in Model 2 given by regression equation (4).

- Recall that the marginal effects on $price_i$ of wgt_i and mpg_i in regression equation (4) are respectively:

$$\frac{\partial \text{price}_i}{\partial \text{wgt}_i} = \beta_1 + 2\beta_3 \text{wgt}_i + \beta_5 \text{mpg}_i \text{ = marginal effect of } \boldsymbol{wgt_i}$$

$$\frac{\partial \text{price}_i}{\partial \text{mpg}_i} = \beta_2 + 2\beta_4 \text{mpg}_i + \beta_5 \text{wgt}_i \text{ = marginal effect of } \boldsymbol{mpg_i}$$

Hypotheses 1 and 2 imply the same coefficient exclusion restriction on regression equation (4), namely the restriction $\beta_5 = 0$.

1. The ***marginal* effect of *wgt_i* on *$price_i$* is unrelated to *mpg_i*** if $\beta_5 = 0$ in regression equation (4).
2. The ***marginal* effect of *mpg_i* on *$price_i$* is unrelated to *wgt_i*** if $\beta_5 = 0$ in regression equation (4).

- The ***null* and *alternative* hypotheses** for both these propositions are:

H_0: $\beta_5 = 0$

H_1: $\beta_5 \neq 0$.

- The following **test** command computes an **F-test** of H_0 against H_1. Enter the commands:

```
test wgtmpg      or    test wgtmpg = 0
return list
```

Inspect the results generated by these commands. State the inference you would draw from this F-test.

- The following **lincom** command computes a ***two-tail* t-test** of H_0 against H_1. Enter the following commands:

```
lincom _b[wgtmpg]
return list
scalar t9 = r(estimate)/r(se)
scalar list t9
```

```
display 2*ttail(68, abs(t9))
display t9^2
display Ftail(1, 68, t9^2)
```

Inspect the results generated by these commands. State the inference you would draw from this t-test.

Remarks: The above **lincom** command simply replicates the two-tail t-test of this same hypothesis that is reported by the **regress** command used to estimate regression equation (4) by OLS. You should be able to explain and verify from the above results why the **two-tail t-test** of H_0: $\beta_5 = 0$ versus H_1: $\beta_5 \neq 0$ ***is equivalent to*** the **F-test** of the same hypothesis.

❑ Higher Order Polynomial Terms in Continuous Explanatory Variables

Before concluding that Model 2 given by regression equation (4) provides an adequate representation of the relationships of wgt_i and mpg_i to North American car prices, we should consider whether higher order polynomial terms in the two continuous explanatory variables wgt_i and mpg_i are necessary to capture these relationships. By higher order terms, we mean the third and fourth powers of the continuous variables wgt_i and mpg_i.

<u>Model 3</u>:

$$\begin{aligned} price_i = \beta_0 + \beta_1 wgt_i + \beta_2 mpg_i + \beta_3 wgt_i^2 + \beta_4 mpg_i^2 + \beta_5 wgt_i mpg_i \\ + \beta_6 wgt_i^3 + \beta_7 mpg_i^3 + \beta_8 wgt_i^4 + \beta_9 mpg_i^4 + u_i \end{aligned} \qquad \textbf{(5)}$$

where

wgt_i^3 = the cubic (third power) of wgt_i for the i-th car;
mpg_i^3 = the cubic (third power) of mpg_i for the i-th car;
wgt_i^4 = the quartic (fourth power) of wgt_i for the i-th car;
mpg_i^4 = the quartic (fourth power) of mpg_i for the i-th car.

The main question we want to address with Model 3 is: Do we need the additional cubic and quartic terms to adequately represent the relationship of $price_i$ to wgt_i and mpg_i, since these additional regressors considerably complicate the marginal effects of the two continuous explanatory variables wgt_i and mpg_i?

To address this question, we test three sets of exclusion restrictions on the regression coefficients of Model 3.

The first test (***Test 10***) addresses the question: Do we need at least one of the four additional regressors that Model 3 adds to Model 2, i.e., at least one of the regressors wgt_i^3, mpg_i^3, wgt_i^4 and mpg_i^4? This test involves jointly testing on Model 3 the four coefficient restrictions $\beta_6 = 0$, $\beta_7 = 0$, $\beta_8 = 0$, and $\beta_9 = 0$.

The second test (***Test 11***) addresses the question: Do we need at least one of the two regressors wgt_i^3 and wgt_i^4 to adequately represent the conditional effect of wgt_i on $price_i$? This test involves jointly testing on Model 3 the two coefficient restrictions $\beta_6 = 0$ and $\beta_8 = 0$.

The third test (***Test 12***) addresses the question: Do we need at least one of the two regressors mpg_i^3 and mpg_i^4 to adequately represent the conditional effect of mpg_i on $price_i$? This test involves jointly testing on Model 3 the two coefficient restrictions $\beta_7 = 0$ and $\beta_9 = 0$.

- To begin, you must create the new variables wgt_i^3, mpg_i^3, wgt_i^4 and mpg_i^4. Enter the following **generate** commands:

```
generate wgt3rd = wgt^3
generate mpg3rd = mpg^3
generate wgt4th = wgt^4
generate mpg4th = mpg^4
```

- Now estimate by OLS Model 3 as given by regression equation (5). Enter the following **regress** command:

```
regress price wgt mpg wgtsq mpgsq wgtmpg wgt3rd mpg3rd wgt4th
mpg4th
```

- ***<u>Test 10</u>:*** Test the hypothesis that all four of the coefficients of the regressors wgt_i^3, mpg_i^3, wgt_i^4 and mpg_i^4 equal zero in Model 3 given by regression equation (5).

- The ***null*** **and** ***alternative*** **hypotheses** for both these propositions are:

  H_0: $\beta_6 = 0$ *and* $\beta_7 = 0$ *and* $\beta_8 = 0$ *and* $\beta_9 = 0$

  H_1: $\beta_6 \neq 0$ *and/or* $\beta_7 \neq 0$ *and/or* $\beta_8 \neq 0$ *and/or* $\beta_9 \neq 0$

- The following **test** command computes an **F-test** of H_0 against H_1. Enter the command:

```
test wgt3rd mpg3rd wgt4th mpg4th
```

  You will immediately notice that the above **test** command has dropped one of the four coefficient exclusion restrictions, specifically the restriction that the slope coefficient of the regressor wgt_i^4 (**`wgt4th`**) equals zero. This has happened because $\hat{\beta}_8$, the OLS slope coefficient estimate for wgt_i^4, is numerically so small (so close to zero) that *Stata* cannot compute the sample value of the required F statistic if the restriction $\beta_8 = 0$ is retained in the null hypothesis.

  The fix for this problem is to **re-scale the variable *wgt_i*** in such a way as to reduce the sample values of the regressors that include wgt_i.

- **Re-scaling the regressors in *wgt_i*:** Recall that the variable ***wgt_i*** **measures car weight in *pounds***. To make the sample values of the variable wgt_i smaller, create a new weight variable by dividing wgt_i by 100; this has the effect of changing the units in which car weight is measured from **pounds** to ***hundreds*** **of pounds**. Thus, define the re-scaled weight variable ***$wgt00_i = wgt_i/100$***, where ***$wgt00_i$*** is the weight of the i-th car measured in ***hundreds*** **of pounds**. You will also need to create new variables for the regressors that are functions of the re-scaled weight variable ***$wgt00_i$***. Enter the following **generate** commands:

```
generate wgt00 = wgt/100
generate wgt00sq = wgt00^2
generate wgt003rd = wgt00^3
generate wgt004th = wgt00^4
```

```
generate wgt00mpg = wgt00*mpg
```

- The revised Model 3 is obtained by simply replacing the original weight variable wgt_i with the re-scaled weight variable $wgt00_i$ in regression equation (5):

Revised Model 3:

$$\begin{aligned} price_i = \ & \beta_0 + \beta_1^* wgt00_i + \beta_2 mpg_i + \beta_3^* wgt00_i^2 + \beta_4 mpg_i^2 + \beta_5^* wgt00_i\, mpg_i \\ & + \beta_6^* wgt00_i^3 + \beta_7 mpg_i^3 + \beta_8^* wgt00_i^4 + \beta_9 mpg_i^4 + u_i \end{aligned} \qquad \textbf{(5*)}$$

where

$wgt00_i$ = the weight of the i-th car (in <u>*hundreds*</u> *of pounds*);
$wgt00_i^2$ = the square (second power) of $wgt00_i$;
$wgt00_i^3$ = the cubic (third power) of $wgt00_i$;
$wgt00_i^4$ = the quartic (fourth power) of $wgt00_i$;
$wgt00_i\, mpg_i$ = the interaction of $wgt00_i$ and mpg_i.

- **Estimate by OLS revised Model 3** as given by regression equation (5*). Enter the following **regress** command:

```
regress price wgt00 mpg wgt00sq mpgsq wgt00mpg wgt003rd mpg3rd
wgt004th mpg4th
```

❑ ***Compare the coefficient estimates of regression equations (5) and (5*)***

Note how the OLS coefficient estimates of the revised Model 3 given by equation (5*) are related to the OLS coefficient estimates of the original Model 3 given by equation (5).

$$\hat{\beta}_1^* = 100\,\hat{\beta}_1$$
$$\hat{\beta}_3^* = 100^2\,\hat{\beta}_3 = 10{,}000\,\hat{\beta}_3$$
$$\hat{\beta}_5^* = 100\,\hat{\beta}_5$$
$$\hat{\beta}_6^* = 100^3\,\hat{\beta}_6 = 1{,}000{,}000\,\hat{\beta}_6$$
$$\hat{\beta}_8^* = 100^4\,\hat{\beta}_8 = 100{,}000{,}000\,\hat{\beta}_8$$

Note too that regression equation (5) containing the explanatory variable $\boldsymbol{wgt_i}$ is **observationally equivalent** to regression equation (5*) containing the re-scaled explanatory variable $\boldsymbol{wgt00_i}$: all summary statistics (the ANOVA F-statistic, the R^2 and adjusted-R^2 values, and the estimated $\hat{\sigma}$ value) are equal for regression equations (5) and (5*), as are the t-ratios and two-tail p-values of the individual coefficient estimates. In other words, apart from re-scaling some of the slope coefficient estimates, changing the units in which car weight is measured (in this case from *pounds* to *hundreds of pounds*) leaves unchanged the basic relationship of car price ($price_i$) to the two continuous explanatory variables car weight (wgt_i or $wgt00_i$) and fuel efficiency (mpg_i).

- ***<u>Test 10 again</u>:*** Using the estimates of regression equation (5*), repeat the test of the hypothesis that all four of the coefficients of the regressors $wgt00_i^3$, mpg_i^3, $wgt00_i^4$ and mpg_i^4 equal zero.

- The ***null* and *alternative* hypotheses** for both these propositions are:

  H_0: $\beta_6^* = 0$ *and* $\beta_7 = 0$ *and* $\beta_8^* = 0$ *and* $\beta_9 = 0$

  H_1: $\beta_6^* \neq 0$ *and/or* $\beta_7 \neq 0$ *and/or* $\beta_8^* \neq 0$ *and/or* $\beta_9 \neq 0$

- The following **test** command computes an **F-test** of H_0 against H_1. Enter the commands:

```
test wgt003rd mpg3rd wgt004th mpg4th
return list
```

The above **test** command correctly computes the required F-statistic. Since that F-statistic has a p-value of 0.0403, the null hypothesis H_0 is rejected at the 5 percent significance level. The implication of this test outcome is that one or more of the slope coefficients β_6^*, β_7, β_8^*, and β_9 is non-zero. In other words, the four coefficient exclusion restrictions that Model 2 imposes on Model 3 are empirically invalid.

- ***Test 11:*** Using the estimates of regression equation (5*), test the hypothesis that the coefficients of both the regressors $wgt00_i^3$ and $wgt00_i^4$ equal zero.

- The ***null* and *alternative* hypotheses** for this proposition are:

H_0: $\beta_6^* = 0$ *and* $\beta_8^* = 0$

H_1: $\beta_6^* \neq 0$ *and/or* $\beta_8^* \neq 0$

- The following **test** command computes an **F-test** of H_0 against H_1. Enter the commands:

```
test wgt003rd wgt004th
return list
```

The above **test** command produces an F-statistic that has a p-value of 0.1972. The null hypothesis H_0 is therefore retained (not rejected) at the 5 percent significance level. The implication of this test outcome is that both slope coefficients β_6^* and β_8^* are equal to zero. In other words, the cubic and quartic terms in ***wgt00**$_i$* are probably not needed to represent adequately the conditional effect of car weight on car price.

- ***Test 12:*** Using the estimates of regression equation (5*), test the hypothesis that the coefficients of both the regressors mpg_i^3 and mpg_i^4 equal zero.

- The ***null* and *alternative* hypotheses** for this proposition are:

H_0: $\beta_7 = 0$ *and* $\beta_9 = 0$

H_1: $\beta_7 \neq 0$ *and/or* $\beta_9 \neq 0$

- The following **test** command computes an **F-test** of H_0 against H_1. Enter the commands:

```
test mpg3rd mpg4th
return list
```

The above **test** command computes an F-statistic that has a p-value of 0.0315. The null hypothesis H_0 is therefore rejected at the 5 percent significance level. This test outcome implies that one or both of the slope coefficients β_7 and β_9 are non-zero. In other words, at least one of the polynomial terms mpg_i^3 and mpg_i^4 is needed in order to adequately represent the conditional relationship of fuel efficiency to car price.

Taken together, the results of hypothesis tests 10, 11 and 12 suggest that Model 3 can perhaps be simplified by dropping from regression equation (5*) the cubic and quartic terms in $wgt00_i$ ($wgt00_i^3$ and $wgt00_i^4$) and the quartic term in mpg_i (mpg_i^4). The following hypothesis test determines whether the omission of these three regressors from regression equation (5*) is consistent with the sample data.

- ***Test 13*:** Using the estimates of regression equation (5*), test the hypothesis that the coefficients of the regressors $wgt00_i^3$, $wgt00_i^4$ and mpg_i^4 are jointly equal to zero.

- The ***null* and *alternative* hypotheses** for this proposition are:

  H_0: $\beta_6^* = 0$ *and* $\beta_8^* = 0$ *and* $\beta_9 = 0$

  H_1: $\beta_6^* \neq 0$ *and/or* $\beta_8^* \neq 0$ *and/or* $\beta_9 \neq 0$

- The following **test** command computes an **F-test** of H_0 against H_1. Enter the commands:

```
test wgt003rd wgt004th mpg4th
return list
```

The above **test** command produces an F-statistic that has a p-value of 0.2543. The null hypothesis H_0 is therefore retained (not rejected) at the 5 percent significance level, and even at the 20 percent significance level. This test outcome implies that regression equation (5*) can be simplified by omitting the regressors $wgt00_i^3$, $wgt00_i^4$ and mpg_i^4, i.e., by imposing on regression equation (5*) the three coefficient exclusion restrictions $\beta_6^* = 0$, $\beta_8^* = 0$ and $\beta_9 = 0$.

<u>Model 4 – a restricted version of Model 3</u>:

$$\text{price}_i = \beta_0 + \beta_1^* \text{wgt00}_i + \beta_2 \text{mpg}_i + \beta_3^* \text{wgt00}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5^* \text{wgt00}_i \text{mpg}_i + \beta_7 \text{mpg}_i^3 + u_i \qquad \textbf{(6)}$$

- **Estimate by OLS Model 4** as given by regression equation (6). Enter the following **regress** command:

```
regress price wgt00 mpg wgt00sq mpgsq wgt00mpg mpg3rd
```

♦ ***<u>Test 14</u>:*** Test the hypothesis that the ***marginal* effect of *wgt00$_i$* on *price$_i$* is *zero*** in Model 4 given by regression equation (6).

- The marginal effect of wgt00_i on price_i is obtained by partially differentiating regression equation (6) with respect to wgt00_i:

$$\frac{\partial\, \text{price}_i}{\partial\, \text{wgt00}_i} = \beta_1^* + 2\beta_3^* \text{wgt00}_i + \beta_5^* \text{mpg}_i .$$

Note that in Model 4, the marginal effect of wgt00_i on price_i is a linear function of both wgt00_i and mpg_i.

- Sufficient conditions for $\partial\, \text{price}_i / \partial\, \text{wgt00}_i = 0$ for all i are $\beta_1^* = 0$ *and* $\beta_3^* = 0$ *and* $\beta_5^* = 0$. We therefore want to test these three coefficient exclusion restrictions.

- The ***null* and *alternative* hypotheses** are:

  H_0: $\beta_1^* = 0$ *and* $\beta_3^* = 0$ *and* $\beta_5^* = 0$

  H_1: $\beta_1^* \neq 0$ *and/or* $\beta_3^* \neq 0$ *and/or* $\beta_5^* \neq 0$.

- The following **test** command computes an F-test of H_0 against H_1. Enter the commands:

```
test wgt00 wgt00sq wgt00mpg
return list
```

Inspect the results generated by these commands. State the inference you would draw from this test.

- ***Test 15:*** Test the hypothesis that the ***marginal*** **effect of** ***mpg_i*** **on** ***$price_i$*** **is** ***zero*** in Model 4 given by regression equation (6).

- The marginal effect of mpg_i on $price_i$ is obtained by partially differentiating regression equation (6) with respect to mpg_i.

$$\frac{\partial \text{price}_i}{\partial \text{mpg}_i} = \beta_2 + 2\beta_4 \text{mpg}_i + \beta_5^* \text{wgt00}_i + 3\beta_7 \text{mpg}_i^2 .$$

Note that in Model 4, the marginal effect of mpg_i on $price_i$ is a linear function of $wgt00_i$ and a quadratic function of mpg_i.

- Sufficient conditions for $\partial \text{price}_i / \partial \text{mpg}_i = 0$ for all i are $\beta_2 = 0$ *and* $\beta_4 = 0$ *and* $\beta_5^* = 0$ *and* $\beta_7 = 0$. We therefore want to test these four coefficient restrictions on regression equation (6).

- The ***null*** **and** ***alternative*** **hypotheses** are:

H_0: $\beta_2 = 0$ *and* $\beta_4 = 0$ *and* $\beta_5^* = 0$ *and* $\beta_7 = 0$

H_1: $\beta_2 \neq 0$ *and/or* $\beta_4 \neq 0$ *and/or* $\beta_5^* \neq 0$ *and/or* $\beta_7 \neq 0$.

- The following **test** command computes an F-test of H_0 against H_1. Enter the commands:

```
test mpg mpgsq wgt00mpg mpg3rd
return list
```

Inspect the results generated by these commands. State the inference you would draw from this test.

Computing Estimates of the Marginal Effects of Continuous Explanatory Variables – Model 4

In applied econometric analyses, one often wants to compute estimates of the magnitude of the marginal effects of those continuous explanatory variables that are of primary interest. This section demonstrates how to do this for Model 4 given by regression equation (6).

Since the marginal effects of both $wgt00_i$ and mpg_i in Model 4 are functions of these two variables, it is necessary to select specific values of $wgt00_i$ and mpg_i at which to estimate their marginal effects. Suppose we decide to estimate the marginal effects of $wgt00_i$ and mpg_i for the median car sold in North America, that is, for the car that has the **sample *median* value of *$wgt00_i$*** and the **sample *median* value of *mpg_i***.

- First, compute the sample median values of the explanatory variables $wgt00_i$ and mpg_i. Enter the following **summarize** commands with the **detail** option:

```
summarize wgt00, detail
summarize mpg, detail
```

  Note that the sample median (50^{th} percentile) value of $wgt00_i$ is 31.9 hundreds of pounds, while the sample median (50^{th} percentile) value of mpg_i is 20 miles per gallon.

- Compute the **estimated value of the marginal effect of *$wgt00_i$*** in equation (6) for a car that weighs 31.9 hundreds of pounds (3,190 pounds) and has fuel efficiency of 20 miles to the gallon. Recall that the expression for $\partial \text{price}_i / \partial \text{wgt00}_i$ in equation (6) is:

$$\frac{\partial \text{price}_i}{\partial \text{wgt00}_i} = \beta_1^* + 2\beta_3^* \text{wgt00}_i + \beta_5^* \text{mpg}_i .$$

  Evaluated at $wgt00_i = 31.9$ and $mpg_i = 20$, the *estimated* marginal effect of $wgt00_i$ on $price_i$ in Model 4 is:

$$\text{estimate of } \frac{\partial \text{price}_i}{\partial \text{wgt00}_i} = \hat{\beta}_1^* + 2\hat{\beta}_3^*(31.9) + \hat{\beta}_5^*(20) .$$

- Use a **lincom** command to compute the estimated value of the marginal effect of $wgt00_i$ in equation (6) for a car that weighs 31.9 hundreds of pounds (3,190 pounds) and has fuel efficiency of 20 miles to the gallon. Enter the commands:

```
lincom _b[wgt00] + 2*_b[wgt00sq]*31.9 + _b[wgt00mpg]*20
return list
```

Inspect the output generated by this **lincom** command. Would you infer that the marginal effect of $wgt00_i$ on $price_i$ is different from zero for a car that weighs 31.9 hundreds of pounds and has fuel efficiency equal to 20 miles per gallon?

- Compute the **estimated value of the marginal effect of *mpg_i*** in equation (6) for a car that weighs 31.9 hundreds of pounds (3,190 pounds) and has fuel efficiency of 20 miles to the gallon. Recall that the expression for $\partial \text{price}_i / \partial \text{mpg}_i$ in equation (6) is:

$$\frac{\partial \text{price}_i}{\partial \text{mpg}_i} = \beta_2 + 2\beta_4 \text{mpg}_i + \beta_5^* \text{wgt00}_i + 3\beta_7 \text{mpg}_i^2 .$$

Evaluated at $wgt00_i = 31.9$ and $mpg_i = 20$, the *estimated* marginal effect of mpg_i on $price_i$ in Model 4 is:

$$\text{estimate of } \frac{\partial \text{price}_i}{\partial \text{mpg}_i} = \hat{\beta}_2 + 2\hat{\beta}_4(20) + \hat{\beta}_5^*(31.9) + 3\hat{\beta}_7(20)^2 .$$

- Use a **lincom** command to compute the estimated value of the marginal effect of mpg_i in equation (6) for a car that weighs 31.9 hundreds of pounds (3,190 pounds) and has fuel efficiency of 20 miles to the gallon. Enter the commands:

```
lincom _b[mpg] + 2*_b[mpgsq]*20 + _b[wgt00mpg]*31.9 +
3*_b[mpg3rd]*20*20
return list
```

Inspect the output generated by this **lincom** command. Would you infer that the marginal effect of mpg_i on $price_i$ is different from zero for a car that weighs 31.9 hundreds of pounds and has fuel efficiency equal to 20 miles per gallon?

❑ Preparing to End Your *Stata* Session

<u>Before you end your *Stata* session</u>, you should do two things.

- First, you will want to **save the current data set**. Enter the following **save** command with the **replace** option to save the current data set as *Stata*-format data set **auto3.dta** and overwrite any existing dataset of the same name:

```
save auto3, replace
```

- Second, **close the log file** you have been recording. Enter the command:

```
log close
```

Alternatively, you could have closed the log file by:

- clicking on the **Log** button in the *Stata* button bar;
- clicking on **<u>C</u>lose log file** in the **Stata Log Options** dialog box;
- clicking the **OK** button.

❑ End Your *Stata* Session – exit

- **<u>To end your *Stata* session</u>**, use the **exit** command. Enter the command:

```
exit        or    exit, clear
```

❑ Cleaning Up and Clearing Out

<u>After returning to Windows</u>, you should copy all the files you have used and created during your *Stata* session to your own portable electronic storage device such as a flash memory stick. These files will be found in the ***Stata working directory***, which is usually **C:\data** on the computers in Dunning 350. There are two files you will want to be sure you have: the *Stata* log file **452tutorial3.log**; and the *Stata*-format data set **auto3.dta** you have saved. Use the Windows **copy** command to copy any files you want to keep to your own portable electronic storage device (e.g., a flash memory stick).

Finally, **<u>as a courtesy to other users</u>** of the computing classroom, please delete all the files you have used or created from the *Stata* working directory.